Použití integrálů

V textu skript jsou jako aplikace integrálu uvedeny výpočty obsahů některých rovinných obrazců, objemy těles, povrch těles a těžiště různých objektů. Většinou se pouze dosazuje do vzorečků, talže Matthematica tu slouží jen pro výpočet příslušného integrálu ze vzorečku. To tu samozřejmě dělat nebudeme. Je možné vyrobit pro příslušné vzorečky novou funkci - to ukážeme např. na délce křivky, ostatní výpočty můžete zadefinovat jako funkce sami. Zbývá graficky vysvětlit některé postupy, které ukazují, jak se ke vzorcům dospělo.

Délka křivky

Pouřijeme vzorec pro výpočet délky grafu funkce f na intervalu [a,b]. Inntegrál ve vzorci se málokdy spočte přesně, proto je lepší použít numerickou integraci. Protože výsledek může být někdy napsán jako komplexní číslo s nulovou imagonární složkou, budeme brát jen reálnou část. Podobně můžete sami sestavit funkce pro další výpočty, např. objemy, délky křiveka zadaných parametricky, atd.

```
Clear[f]; f[x_] = Input["Napište funkci proměnné x"];
s = ToExpression[
   StringSplit[
    InputString[
     "Zadejte interval, na kterém počítáte délku
       grafu , ve tvaru levý krajmí bod pravý
       krajní bod"]]];
d = Integrate[Sqrt[1 + (D[f[x], x])^2],
    {x, s[[1]] , s[[2]]}] // N;
Row[{"Délka grafu funkce ", f[x], " na intervalu [" ,
  s[[1]] , ",", s[[2]] , "] je rovna ", Re[d]}]
```

Délka grafu funkce cos(x) na intervalu [0,2] je rovna 2.50798

Jinou možností je přímo zadefinovat délku jako novou funkci.

```
delka[f_, a_, b_] :=
 Re[Integrate[Sqrt[1 + (D[f[x], x])^2], {x, a , b} ] // N]
```

```
delka[Cos, 0, 2]
```

2.50798

V této definici nelze použít přímo za f např. funkci x^3. Je nutné nejdříve funkci zadefinovat a použít jen jení název.

```
delka[x^3, 0, 3]
```

— *NIntegrate::inumr :*

The integrand $\sqrt{1+\left(\left(3\,x^2\right)(x)+\left(x^3\right)'(x)\right)^2}$ *has evaluated to non–numerical values for all sampling points in the region with boundaries* $(0\quad 3)$. ≫

— *NIntegrate::inumr :*

The integrand $\sqrt{1+\left(\left(3\,x^2\right)(x)+\left(x^3\right)'(x)\right)^2}$ *has evaluated to non–numerical values for all sampling points in the region with boundaries* $(0\quad 3)$. ≫

— *NIntegrate::inumr :*

The integrand $\sqrt{1+\left(\left(3\,x^2\right)(x)+\left(x^3\right)'(x)\right)^2}$ *has evaluated to non–numerical values for all sampling points in the region with boundaries* $(0\quad 3)$. ≫

— *General::stop :*

Further output of NIntegrate::inumr will be suppressed during this calculation. ≫

$$\mathrm{Re}\left(\mathrm{NIntegrate}\left[\sqrt{\left(\left(x^3\right)'(x)+\left(3\,x^2\right)(x)\right)^2+1}\,,\ \{x,\ 0,\ 3\}\right]\right)$$

```
g[x_] := x^3
```

```
delka[g, 0, 3]
```

27.6581

Délka křivky se definuje jako supremum délek lomených čar, které začínají a končí ve stejných bodech má daná křivka a body lomu
leží na křivce. Následující animace ukazuje, jak se u hezké křivky uvedené lomené čáry a jejich délky blíží ke křivce a její délce. Křivkou v ukázce bude půlkružnice, ale snadno můžete v zadání

změnit interval i funkci.

```
Clear[f, a, b]; f[x_] := Sqrt[1 - x^2]; a = -1; b = 1;
```

```
bodylomu[n_] := Table[{k, f[k]}, {k, a, b, (b - a) / n}];
delkalomcary[n_] :=
 Sum[Sqrt[((b - a) / n)^2 + (f[k + (b - a) / n] - f[k])^2] // N,
  {k, a, (a + (n - 1) b) / n, (b - a) / n}];
delkakrivky = Integrate[Sqrt[1 + (D[f[x], x])^2], {x, a, b}];
```

```
Manipulate[
 Show[{Plot[f[x], {x, a, b}, PlotStyle → Blue,
    PlotRange → {0, 1.4}],
   ListLinePlot[bodylomu[n], PlotStyle → Red],
   Graphics[{Text["Délka křivky =", {.1, 1.1}, {-1, 0}],
     Text[delkakrivky // N, {.57, 1.1}, {-1, 0}],
     Text["Délka lomené čáry =", {0.1, 1.3}, {-1, 0}],
     Text[delkalomcary[n], {.7, 1.3}, {-1, 0}]}]
   ]}], {n, 1, 32, 1}]
```